MATHEMATICS OF COMPUTATION
VOLUME 49, NUMBER 180
OCTOBER 1987, PAGES 595-606

A Table of Elliptic Integrals of the Second Kind*

By B. C. Carlson

Abstract. By evaluating elliptic integrals in terms of standard R-functions instead of Legendre's integrals, many (in one case 144) formulas in previous tables are unified. The present table includes only integrals of the first and second kinds having integrands with real singular points. The 216 integrals of this type listed in Gradshteyn and Ryzhik's table constitute a small fraction of the special cases of 13 integrals evaluated here. The interval of integration is not required, as it is in previous tables, to begin or end at a singular point of the integrand. Fortran codes for the standard R-functions are included in a Supplement.

1. Introduction. Let

$$(1.1)\qquad [p] = [p_1, p_2, \dots, p_n] = \int_y^x (a_1 + b_1 t)^{p_1/2} \cdots (a_n + b_n t)^{p_n/2}\, dt,$$

where $p_1, \dots, p_n$ are nonzero integers, the integrand is real, and the integral is assumed to be well defined. Many integrals like

$$\int (1 - k^2 \sin^2\phi)^{p_1/2}\, d\phi \quad \text{and} \quad \int (a + bz^2)^{p_1/2}(c + dz^2)^{p_2/2}\, dz$$

can be put in the form (1.1) by letting $t = \sin^2\phi$ or $t = z^2$.

For purposes of classification we assume the b's are nonzero and no two of the quantities $a_i + b_i t$ are proportional. If at most two p's are odd, the integral (1.1) is elementary. If exactly three p's are odd (the "cubic case"), the integral is elliptic of the first or second kind if all the even p's are positive, and otherwise it is third kind. The only such integral of the first kind is $[-1, -1, -1]$. If exactly four p's are odd (the "quartic case"), the integral is elliptic of the first or second kind if all the even p's are positive and $p_1 + \cdots + p_n \leqslant -4$; otherwise it is third kind. The only such integral of the first kind is $[-1, -1, -1, -1]$. If more than four p's are odd, the integral is hyperelliptic.

Integrals of the first kind are traditionally expressed in terms of Legendre's $F(\phi, k)$ with $0 \leqslant k \leqslant 1$ and $0 \leqslant \phi \leqslant \pi/2$. Integrals of the second kind require

Received October 6, 1986.
1980 *Mathematics Subject Classification* (1985 *Revision*). Primary 33A25; Secondary 33A30.
*Part of this work was done at the University of Maryland, where the author was a visitor at the Institute for Physical Science and Technology, with the support of AROD contract DAAG 29-80-C-0032. The rest was done in the Ames Laboratory, which is operated for the U. S. Department of Energy by Iowa State University under contract no. W-7405-ENG-82. The work was supported by the Director of Energy Research, Office of Basic Energy Sciences.

©1987 American Mathematical Society
0025-5718/87 $1.00 + $.25 per page

$E(\phi, k)$ and usually F also. We shall replace F by the symmetric integral

$$R_F(x, y, z) = \frac{1}{2}\int_0^{\infty} [(t + x)(t + y)(t + z)]^{-1/2}\, dt \tag{1.2}$$

and E by

$$R_D(x, y, z) = \frac{3}{2}\int_0^{\infty} (t + x)^{-1/2}(t + y)^{-1/2}(t + z)^{-3/2}\, dt. \tag{1.3}$$

These R-functions are homogeneous:

$$\begin{aligned} R_F(\lambda x, \lambda y, \lambda z) &= \lambda^{-1/2} R_F(x, y, z), \\ R_D(\lambda x, \lambda y, \lambda z) &= \lambda^{-3/2} R_D(x, y, z), \end{aligned} \tag{1.4}$$

and they are normalized so that

$$R_F(x, x, x) = x^{-1/2}, \qquad R_D(x, x, x) = x^{-3/2}. \tag{1.5}$$

Fortran codes [6] for computing R_F and R_D when x, y, z are real and nonnegative are listed in the Supplements section of this issue and can be found also in most of the major software libraries.

Customary integral tables [1], [7], [9] assume that the interval of integration begins or ends at a branch point of the integrand, and many special cases are listed according to the positions of the other branch points relative to the interval of integration and to one another. If the integral at hand does not have either limit of integration at a branch point, it must be split into two parts that do. In the present paper these two parts are recombined by the addition theorem, and the need to specify the relative positions of the branch points then disappears. The use of R-functions greatly facilitates the application of the addition theorem and leads to a further unification that cannot be achieved with Legendre's integrals, because the expressions for $R_F(x, y, z)$ and $R_D(x, y, z)$ in terms of Legendre's integrals with $0 \leqslant k \leqslant 1$ and $0 \leqslant \phi \leqslant \pi/2$ depend on the relative sizes of x, y, and z (see [5, (4.1), (4.2)], (5.25), and (5.32)).

Integrals of the third kind and integrands with conjugate complex branch points, resulting from an irreducible quadratic factor $a_i + b_i t + c_i t^2$, will be deferred to later papers. (Integrals of the first kind with quadratic factors are treated in [3].) The main table in Section 2 consists of quartic cases, since cubic cases can be obtained from these by putting $a_i = 1$ and $b_i = 0$ for various choices of i. To select integrals that are relatively simple and occur most commonly in practice, we arbitrarily require $\Sigma|p_i| \leqslant 8$. Apart from permutation of subscripts in (1.1), there are just nine quartic cases of the first or second kind satisfying this criterion: $[-1, -1, -1, -1]$, $[1, -1, -1, -3]$, $[-1, -1, -1, -3, 2]$, $[-1, -1, -3, -3]$, $[1, -1, -3, -3]$, $[1, 1, -3, -3]$, $[-1, -1, -1, -5]$, $[1, -1, -1, -5]$, and $[1, 1, -1, -5]$. The integral $[-1, -1, -1, -3]$ is a special case of $[-1, -1, -1, -3, 2]$ with $a_5 = 1$ and $b_5 = 0$.

Section 3 presents four cubic cases not contained in the nine quartic cases: $[3, -1, -3]$, $[3, -1, -1]$, $[-3, -3, -3]$, and $[1, 1, 1]$.

The method of evaluating the integrals is discussed in Sections 4 and 5. The fundamental integrals are $[-1, -1, -1, -1]$ and $[1, -1, -1, -3]$, and the rest are obtained from these by recurrence relations. The single formula (2.7) for $[1, -1, -1, -3]$ replaces 72 cases occupying the nine pages of §3.168 in Gradshteyn and Ryzhik's table [7], as well as 72 cubic cases: 18 cases of $[-1,-1,-3]$ in §3.133, 18 cases of $[1,-1,-1]$ in §3.141, and 36 cases of $[1,-1,-3]$ in §3.142.

By using [5, (4.1), (4.2)], (2.6) was checked against formulas 1, 3, 5, 7 of [7, §3.147], and (2.7) was checked against formulas 1, 5, 42, 70 of [7, §3.168]. The nine integrals in Section 2 and the four in Section 3 were checked numerically to 6S for $y = 0.5$, $x = 2.0$, $a_i = 0.5 + i$, $b_i = 2.5 - i$ by the SLATEC numerical quadrature routine QNG and the routines for R_F and R_D in the Supplements section of this issue.

2. Table of Quartic Cases. We assume $x > y$ and $a_i + b_i t > 0$, $y < t < x$, for all i, and we define

$$d_{ij} = a_i b_j - a_j b_i, \tag{2.1}$$

$$X_i = (a_i + b_i x)^{1/2}, \qquad Y_i = (a_i + b_i y)^{1/2}, \tag{2.2}$$

$$U_{ij} = (X_i X_j Y_k Y_m + Y_i Y_j X_k X_m)/(x - y), \tag{2.3}$$

where i, j, k, m is any permutation of $1, 2, 3, 4$. These definitions imply

$$U_{ik}^2 - U_{im}^2 = d_{ij} d_{km}, \tag{2.4}$$

and consequently the arguments of the R-functions appearing in the table differ by quantities independent of x and y. If one limit of integration is infinite, (2.3) simplifies to

$$\begin{aligned} U_{ij} &= (b_i b_j)^{1/2} Y_k Y_m + Y_i Y_j (b_k b_m)^{1/2}, \qquad & x = +\infty, \\ U_{ij} &= X_i X_j (b_k b_m)^{1/2} + (b_i b_j)^{1/2} X_k X_m, \qquad & y = -\infty, \end{aligned} \tag{2.5}$$

all square roots being nonnegative.

If one limit of integration is a branch point of the integrand, then X_i or Y_i will be 0 for some value of i (with $p_i \geqslant -1$ since we assume that the integral exists), and one of the two terms in every U_{ij} will vanish. If both limits of integration are branch points, the elliptic integral is called complete, and one of the U_{ij} will be 0. It is not assumed that $b_i \neq 0$ nor that $d_{ij} \neq 0$ unless one of these quantities occurs in a denominator. The relation $d_{ij} = 0$ is equivalent to proportionality of $a_i + b_i t$ and $a_j + b_j t$. The nine quartic cases listed in Section 1 follow. Only the first two are treated by Gradshteyn and Ryzhik [7, §3.147, §3.168].

$$\begin{aligned} &\int_y^x [(a_1 + b_1 t)(a_2 + b_2 t)(a_3 + b_3 t)(a_4 + b_4 t)]^{-1/2}\, dt \\ &\qquad = 2R_F(U_{12}^2, U_{13}^2, U_{14}^2). \end{aligned} \tag{2.6}$$

$$\begin{aligned} &\int_y^x (a_1 + b_1 t)^{1/2} [(a_2 + b_2 t)(a_3 + b_3 t)]^{-1/2} (a_4 + b_4 t)^{-3/2}\, dt \\ &\qquad = \frac{2}{3} d_{12} d_{13} R_D(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{2 X_1 Y_1}{X_4 Y_4 U_{14}}. \end{aligned} \tag{2.7}$$

The next equation remains valid even if $a_5 + b_5 t$ changes sign in the interval of integration.

$$\int_y^x [(a_1 + b_1 t)(a_2 + b_2 t)(a_3 + b_3 t)]^{-1/2}(a_4 + b_4 t)^{-3/2}(a_5 + b_5 t)\, dt$$

$$(2.8) \qquad = \frac{2d_{12}d_{13}d_{54}}{3d_{14}} R_D(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{2d_{15}}{d_{14}} R_F(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{2d_{54}X_1Y_1}{d_{14}X_4Y_4U_{14}}.$$

$$\int_y^x [(a_1 + b_1 t)(a_2 + b_2 t)]^{-1/2}[(a_3 + b_3 t)(a_4 + b_4 t)]^{-3/2}\, dt$$

$$(2.9) \qquad = \frac{2}{3d_{34}^2}(b_3^2 d_{14}d_{24} + b_4^2 d_{13}d_{23}) R_D(U_{13}^2, U_{14}^2, U_{12}^2) - \frac{4b_3b_4}{d_{34}^2} R_F(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{2}{d_{34}^2 U_{12}}\left(\frac{b_3^2 X_4Y_4}{X_3Y_3} + \frac{b_4^2 X_3Y_3}{X_4Y_4}\right).$$

$$\int_y^x (a_1 + b_1 t)^{1/2}(a_2 + b_2 t)^{-1/2}[(a_3 + b_3 t)(a_4 + b_4 t)]^{-3/2}\, dt$$

$$(2.10) \qquad = \frac{2d_{13}d_{14}}{3d_{34}^2}(b_3 d_{24} + b_4 d_{23}) R_D(U_{13}^2, U_{14}^2, U_{12}^2) - \frac{2}{d_{34}^2}(b_3 d_{14} + b_4 d_{13}) R_F(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{2b_3 d_{13} X_4Y_4}{d_{34}^2 U_{12} X_3Y_3} + \frac{2b_4 d_{14} X_3Y_3}{d_{34}^2 U_{12} X_4Y_4}.$$

$$\int_y^x [(a_1 + b_1 t)(a_2 + b_2 t)]^{1/2}[(a_3 + b_3 t)(a_4 + b_4 t)]^{-3/2}\, dt$$

$$(2.11) \qquad = \frac{4d_{12}d_{13}d_{24}}{3d_{34}} R_D(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{2d_{12}}{d_{34}} R_F(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{2}{d_{34}U_{14}}\left(\frac{d_{24}X_1Y_1}{X_4Y_4} - \frac{d_{13}X_2Y_2}{X_3Y_3}\right).$$

$$\int_y^x [(a_1 + b_1 t)(a_2 + b_2 t)(a_3 + b_3 t)]^{-1/2}(a_4 + b_4 t)^{-5/2}\, dt$$

$$(2.12) \qquad = \frac{-4b_4}{9d_{14}}\left(\frac{b_1}{d_{14}} + \frac{b_2}{d_{24}} + \frac{b_3}{d_{34}}\right)\left\{d_{12}d_{13} R_D(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{3X_1Y_1}{X_4Y_4U_{14}}\right\} + \frac{2}{3}\left(\frac{2b_1^2}{d_{14}^2} + \frac{b_1b_2}{d_{14}d_{24}} + \frac{b_1b_3}{d_{14}d_{34}} - \frac{b_2b_3}{d_{24}d_{34}}\right) R_F(U_{12}^2, U_{13}^2, U_{14}^2) - \frac{2b_4^2}{3d_{14}d_{24}d_{34}}(X_1X_2X_3X_4^{-3} - Y_1Y_2Y_3Y_4^{-3}).$$

$$\int_y^x (a_1 + b_1 t)^{1/2}[(a_2 + b_2 t)(a_3 + b_3 t)]^{-1/2}(a_4 + b_4 t)^{-5/2}\,dt$$

$$(2.13)\quad \begin{aligned} &= \frac{2}{9}\left(\frac{b_1}{d_{14}} - \frac{2b_2}{d_{24}} - \frac{2b_3}{d_{34}}\right)\left\{d_{12}d_{13}R_D(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{3X_1Y_1}{X_4Y_4U_{14}}\right\} \\ &\quad - \frac{2b_4 d_{12}d_{13}}{3d_{14}d_{24}d_{34}} R_F(U_{12}^2, U_{13}^2, U_{14}^2) \\ &\quad - \frac{2b_4}{3d_{24}d_{34}}(X_1X_2X_3X_4^{-3} - Y_1Y_2Y_3Y_4^{-3}). \end{aligned}$$

$$\int_y^x [(a_1 + b_1 t)(a_2 + b_2 t)]^{1/2}(a_3 + b_3 t)^{-1/2}(a_4 + b_4 t)^{-5/2}\,dt$$

$$(2.14)\quad \begin{aligned} &= \frac{-2}{9d_{14}d_{34}}(d_{13}d_{24} + d_{23}d_{14})\left\{d_{12}d_{13}R_D(U_{12}^2, U_{13}^2, U_{14}^2) + \frac{3X_1Y_1}{X_4Y_4U_{14}}\right\} \\ &\quad - \frac{2d_{12}d_{13}}{3d_{14}d_{34}} R_F(U_{12}^2, U_{13}^2, U_{14}^2) \\ &\quad - \frac{2}{3d_{34}}(X_1X_2X_3X_4^{-3} - Y_1Y_2Y_3Y_4^{-3}). \end{aligned}$$

3. Cubic Cases. By putting $a_i = 1$ and $b_i = 0$ for various choices of i, 13 cubic cases can be evaluated from the quartic cases in Section 2 and do not need to be listed separately. Eight of these are given by Gradshteyn and Ryzhik [7, §§3.131–3.135, 3.141, 3.142]: [−1, −1, −1], [−1, −1, −1, 2], [−1, −1, −3], [−1, −1, −5], [−1, −3, −3], [1, −1, −1], [1, 1, −1], and [1, −1, −3]. They do not give the other five: [1, 1, −3], [1, 1, −5], [1, −1, −5], [1, −3, −3], and [−1, −1, −3, 2].

In this section we list four cubic cases not contained in the quartic cases of Section 2: [3, −1, −3], [3, −1, −1], [−3, −3, −3], and [1, 1, 1]. Only [−3, −3, −3] is given by Gradshteyn and Ryzhik [7, §3.136], and only two cases of this are listed, each with an infinite limit of integration, because the integral diverges if it begins or ends at a finite branch point with $p_i = -3$. If the closed interval of integration lies in the open interval between two finite branch points with $p_i = -3$, there is no way to evaluate the integral by using previous tables.

In place of (2.3) we define

$$(3.1)\qquad U_i = (X_iY_jY_k + Y_iX_jX_k)/(x - y),$$

where i, j, k is any permutation of 1, 2, 3. Since this implies

$$(3.2)\qquad U_i^2 - U_j^2 = b_k d_{ij},$$

the arguments of the R-functions in the table differ by quantities independent of x and y. If one limit of integration is infinite, (3.1) simplifies to

$$(3.3)\qquad U_i = (b_jb_k)^{1/2}Y_i \quad \text{if } x = +\infty, \qquad U_i = (b_jb_k)^{1/2}X_i \quad \text{if } y = -\infty.$$

the square roots being nonnegative. The remarks in the paragraph preceding (2.6) apply, after replacement of U_{ij} by U_i, also to the following integrals.

$$\int_y^x (a_1+b_1t)^{3/2}(a_2+b_2t)^{-1/2}(a_3+b_3t)^{-3/2}\,dt$$

$$\tag{3.4} = \frac{2d_{13}}{b_3d_{23}}\left\{\frac{1}{3}d_{12}(b_1d_{23}+b_2d_{13})R_D(U_2^2,U_3^2,U_1^2) - d_{12}R_F(U_1^2,U_2^2,U_3^2) + \frac{d_{13}X_2Y_2}{X_3Y_3U_1}\right\} + \frac{2b_1X_1Y_1}{b_3U_1}.$$

$$\int_y^x (a_1+b_1t)^{3/2}[(a_2+b_2t)(a_3+b_3t)]^{-1/2}\,dt$$

$$\tag{3.5} = \frac{4}{3b_2b_3}(b_2d_{13}+b_3d_{12})\left\{\frac{1}{3}d_{12}d_{13}R_D(U_2^2,U_3^2,U_1^2) + \frac{X_1Y_1}{U_1}\right\} - \frac{2d_{12}d_{13}}{3b_2b_3}R_F(U_1^2,U_2^2,U_3^2) + \frac{2b_1}{3b_2b_3}(X_1X_2X_3 - Y_1Y_2Y_3).$$

$$\int_y^x [(a_1+b_1t)(a_2+b_2t)(a_3+b_3t)]^{-3/2}\,dt$$

$$\tag{3.6} \begin{aligned} &= \frac{4b_1b_2}{3d_{12}}\left(\frac{b_1b_2}{d_{12}}+\frac{b_2b_3}{d_{23}}+\frac{b_3b_1}{d_{31}}\right)R_D(U_1^2,U_2^2,U_3^2) \\ &\quad + \frac{2b_1b_2}{d_{12}^2}\left(\frac{b_1}{d_{13}}+\frac{b_2}{d_{23}}\right)R_F(U_1^2,U_2^2,U_3^2) + \frac{2b_3^2}{d_{13}d_{23}X_3Y_3U_3} \\ &\quad - \frac{2}{d_{12}^2U_3}\left(\frac{b_1^3X_2Y_2}{d_{13}X_1Y_1}+\frac{b_2^3X_1Y_1}{d_{23}X_2Y_2}\right) \\ &= \sum \frac{2b_3^2}{d_{13}d_{23}}\left\{\frac{1}{3}b_1b_2R_D(U_1^2,U_2^2,U_3^2) + \frac{1}{X_3Y_3U_3}\right\}, \end{aligned}$$

where Σ denotes summation over cyclic permutations of the subscripts $1,2,3$. The same notation is used in the next formula.

$$\int_y^x [(a_1+b_1t)(a_2+b_2t)(a_3+b_3t)]^{1/2}\,dt$$

$$\tag{3.7} \begin{aligned} &= \frac{-2(\Sigma b_1^2d_{23}^2)}{15b_1b_2b_3^2}\left\{\frac{1}{3}d_{13}d_{23}R_D(U_1^2,U_2^2,U_3^2) + \frac{X_3Y_3}{U_3}\right\} \\ &\quad - \frac{2d_{13}d_{23}}{15b_1b_2b_3^2}(b_1d_{23}+b_2d_{13})R_F(U_1^2,U_2^2,U_3^2) \\ &\quad + \frac{2}{15}X_1X_2X_3\left(\frac{X_1^2}{b_1}+\frac{X_2^2}{b_2}+\frac{X_3^2}{b_3}\right) - \frac{2}{15}Y_1Y_2Y_3\left(\frac{Y_1^2}{b_1}+\frac{Y_2^2}{b_2}+\frac{Y_3^2}{b_3}\right) \\ &= \frac{2}{15}\Sigma\left\{\frac{-d_{12}^2}{b_1b_2}\left[\frac{1}{3}d_{13}d_{23}R_D(U_1^2,U_2^2,U_3^2) + \frac{X_3Y_3}{U_3}\right] + b_1^{-1}(X_1^3X_2X_3 - Y_1^3Y_2Y_3)\right\}. \end{aligned}$$

4. The Two Fundamental Integrals. In this section we shall prove (2.6) and (2.7) for $[-1,-1,-1,-1]$ and $[1,-1,-1,-3]$, from which the remaining integrals can be obtained by the recurrence relations of Section 5. In order that the first part of the proof shall apply for future purposes to $[1,-1,-1,-1,-2]$, which is an integral of the third kind, we do not restrict the number n of factors in (1.1) to be 4. It will be important that these three integrals have $p_1 > -2$ and $\sum p_i = -4$.

In (1.1) we assume $x > y$ and $a_i + b_i t > 0$, $y < t < x$, for all i. In the notation of Section 2 this implies $X_i^2 \geqslant 0$ and $Y_i^2 \geqslant 0$ for all i. Temporarily we assume further that $-a_1/b_1 > x$ and that $a_i + b_i t > 0$, $y \leqslant t \leqslant -a_1/b_1$, for $i > 1$. This assumption, which will later be removed by analytic continuation, means that $-a_1/b_1$ is the first singularity encountered to the right of the interval of integration. The first part of the assumption implies $(a_1 + b_1 x)/b_1 < 0$, whence $X_1^2 > 0$, $b_1 < 0$, and $Y_1^2 > 0$, since $Y_1^2 = X_1^2 - b_1(x-y)$. The second part of the assumption implies $a_i + b_i(-a_1/b_1) > 0$, whence $d_{1i} > 0$, $i > 1$.

We can now split (1.1) into two parts, both well defined if $p_1 > -2$:

$$(4.1)\qquad \begin{aligned}[p] &= \int_y^{-a_1/b_1} \prod_{i=1}^{n} (a_i + b_i t)^{p_i/2}\, dt \\ &\quad - \int_x^{-a_1/b_1} \prod_{i=1}^{n} (a_i + b_i t)^{p_i/2}\, dt = I_y - I_x.\end{aligned}$$

It suffices to consider I_y because I_x is the same with y replaced by x. The interval of integration is mapped onto the positive real line by a change of integration variable:

$$(4.2)\qquad \begin{aligned} u &= \frac{t-y}{Y_1^2(a_1+b_1 t)}, & t &= \frac{y + a_1 Y_1^2 u}{1 - b_1 Y_1^2 u}, \\ \frac{dt}{du} &= \frac{Y_1^4}{\left(1 - b_1 Y_1^2 u\right)^2}, & a_i + b_i t &= \frac{Y_1^2 d_{1i} u + Y_i^2}{1 - b_1 Y_1^2 u},\end{aligned}$$

where $d_{11} = 0$. If $\sum p_i = -4$ the powers of $1 - b_1 Y_1^2 u$ cancel, and we find

$$(4.3)\qquad \begin{aligned} I_y &= Y_1^{4+p_1} \int_0^\infty \prod_{i=2}^{n} \left(Y_1^2 d_{1i} u + Y_i^2\right)^{p_i/2} du \\ &= \prod_{j=2}^{n} (d_{1j})^{p_j/2} \int_0^\infty \prod_{i=2}^{n} \left(u + Y_i^2/Y_1^2 d_{1i}\right)^{p_i/2} du.\end{aligned}$$

The integral I_x is the same with $Y_i^2/Y_1^2 d_{1i}$ replaced by $X_i^2/X_1^2 d_{1i}$, and the difference,

$$X_i^2/X_1^2 d_{1i} - Y_i^2/Y_1^2 d_{1i} = (x-y)/X_1^2 Y_1^2,$$

is positive and independent of i. Using the notation

$$(4.4)\qquad \lambda = (x-y)/X_1^2 Y_1^2, \quad z_i = Y_i^2/Y_1^2 d_{1i}, \quad z_i + \lambda = X_i^2/X_1^2 d_{1i},$$

we find from (4.1), (4.3), (1.2), and (1.3) that

$$(4.5)\qquad \begin{aligned}[-1,-1,-1,-1] &= 2(d_{12}d_{13}d_{14})^{-1/2} \\ &\quad \cdot \left[R_F(z_2, z_3, z_4) - R_F(z_2+\lambda, z_3+\lambda, z_4+\lambda)\right],\end{aligned}$$

$$(4.6)\qquad \begin{aligned}[1,-1,-1,-3] &= \frac{2}{3}(d_{12}d_{13})^{-1/2}(d_{14})^{-3/2} \\ &\quad \cdot \left[R_D(z_2, z_3, z_4) - R_D(z_2+\lambda, z_3+\lambda, z_4+\lambda)\right].\end{aligned}$$

The addition theorem [4, (9), (13)] for R_F is

$$\begin{aligned}(4.7)\qquad R_F(z_2, z_3, z_4) &= R_F(z_2+\lambda, z_3+\lambda, z_4+\lambda)\\ &\quad + R_F(z_2+\mu, z_3+\mu, z_4+\mu),\end{aligned}$$

$$z_i + \mu = \lambda^{-2}\left\{\left[(z_i+\lambda)z_j z_k\right]^{1/2} + \left[z_i(z_j+\lambda)(z_k+\lambda)\right]^{1/2}\right\}^2,$$

where i, j, k is any permutation of 2, 3, 4. Thus (4.5) becomes

$$(4.8)\qquad [-1,-1,-1,-1] = 2(d_{12}d_{13}d_{14})^{-1/2} R_F(z_2+\mu, z_3+\mu, z_4+\mu),$$

$$(4.9)\qquad z_i + \mu = \frac{(X_1X_iY_jY_k + Y_1Y_iX_jX_k)^2}{d_{1i}d_{1j}d_{1k}(x-y)^2} = \frac{U_{1i}^2}{d_{12}d_{13}d_{14}}.$$

By the homogeneity property (1.4) we find

$$(4.10)\qquad [-1,-1,-1,-1] = 2R_F(U_{12}^2, U_{13}^2, U_{14}^2),$$

which is the same as (2.6).

This removal of the d's from the arguments of R_F is the critical step. As shown by (1.2), an argument of R_F must not be negative, and so the functions on the right-hand side of (4.5) require the branch points to be ordered so that d_{12}, d_{13}, and d_{14} are positive. To show that (4.10) holds without the assumption that $-a_1/b_1$ is the first singularity to the right of the interval of integration, we use analytic continuation in b_1 or more conveniently in w, where

$$(4.11)\qquad \begin{aligned} w &= X_1^2 = a_1 + b_1 x, & b_1 &= \frac{w - Y_1^2}{x-y},\\ a_1 &= \frac{xY_1^2 - yw}{x-y}, & \frac{-a_1}{b_1} &= x + \frac{w(x-y)}{Y_1^2 - w}.\end{aligned}$$

We fix x, y, $Y_i > 0$, $1 \leqslant i \leqslant n$, and $X_i > 0$, $2 \leqslant i \leqslant n$. Then a_1 and b_1 are functions of w, and we can make $-a_1/b_1$ be the first singularity to the right of the interval of integration by choosing w positive and sufficiently small. For such values of w we have proved that (4.10) is true. We shall show that both sides of (4.10) are analytic in w on the complex plane cut along the nonpositive real axis. It follows by the permanence of functional relations that (4.10) holds in the cut plane and in particular for all positive values of w. Therefore it holds for any real value of $-a_1/b_1$ outside the closed interval of integration. The last statement is immediately evident from the graph of $a_1 + b_1 t$ as a function of t, since $a_1 + b_1 y$ has been fixed and $w = a_1 + b_1 x$.

To prove analyticity, we recall that an R-function is analytic when each of its arguments lies in the plane cut along the nonpositive real axis [2, (6.8–6), Theorem (6.8–1)]. Since (2.3) shows that $U_{ij} = \alpha_{ij}w^{1/2} + \beta_{ij}$, where α_{ij} and β_{ij} are positive, U_{ij}^2 lies in the cut plane when w does, and so the right-hand side of (4.10) is analytic in the cut w-plane. The left side is defined by (1.1), which can be rewritten, when $\sum p_i = -4$, as

$$(4.12)\qquad [p] = (x-y)\left(\prod_{i=1}^{n} Y_i^{p_i}\right) R_{-1}\left(\frac{-p_1}{2}, \ldots, \frac{-p_n}{2}; \frac{X_1^2}{Y_1^2}, \ldots, \frac{X_n^2}{Y_n^2}\right)$$

by taking $s = (x - t)/(t - y)$ as a new variable of integration and using [2, (6.8–6)]. Since Y_1^2 is positive and $X_1^2 = w$, the right side of (4.12) and the left side of (4.10) are analytic in the cut w-plane, and the proof of (2.6) is complete.

A different proof of (2.6) was given in [3], but the present proof is adaptable to (2.7) with only minor changes. The addition theorem for R_D, obtained by putting $\rho = z$ in [11, (8.11)], is

$$R_D(z_2, z_3, z_4) = R_D(z_2 + \lambda, z_3 + \lambda, z_4 + \lambda) + R_D(z_2 + \mu, z_3 + \mu, z_4 + \mu) + 3[z_4(z_4 + \lambda)(z_4 + \mu)]^{-1/2}, \tag{4.13}$$

where μ is the same as in (4.7). Thus (4.6) becomes

$$[1, -1, -1, -3] = \frac{2}{3}(d_{12}d_{13})^{-1/2}(d_{14})^{-3/2} \cdot \left\{ R_D(z_2 + \mu, z_3 + \mu, z_4 + \mu) + 3[z_4(z_4 + \lambda)(z_4 + \mu)]^{-1/2} \right\}. \tag{4.14}$$

Substituting (4.4) and (4.9) and using the homogeneity property (1.4), we find (2.7). The temporary assumption about $-a_1/b_1$ can again be removed by the permanence of functional relations. In the first term on the right-hand side of (2.7), d_{12} and d_{13} are linear functions of $w = X_1^2$ by (2.1) and (4.11), and R_D is analytic in the cut w-plane by the same reasoning that applied earlier to R_F. The second term also is analytic because $X_1/U_{14} = w^{1/2}/(\alpha_{14}w^{1/2} + \beta_{14})$, where α_{14} and β_{14} are positive. Since the left side of (2.7) is a special case of (4.12), the proof is complete.

5. Recurrence Relations. Let e_i denote an n-tuple with 1 in the ith place and 0's elsewhere (for example, $[p + 2e_1] = [p_1 + 2, p_2, \ldots, p_n]$). We shall first list some relations between different integrals, then give their proofs, and finally show how they can be used to obtain all the integrals in the table from the two fundamental integrals (2.6) and (2.7). The most useful relation is

$$d_{ij}[p] = b_j[p + 2e_i] - b_i[p + 2e_j]. \tag{5.1}$$

Two others, involving the quantity

$$A(p) = \prod_{i=1}^{n} X_i^{p_i} - \prod_{i=1}^{n} Y_i^{p_i}, \tag{5.2}$$

are

$$\sum_{i=1}^{n} p_i b_i[p - 2e_i] = 2A(p) \tag{5.3}$$

and

$$(p_1 + \cdots + p_n + 2)b_i[p] = \sum_{j=1}^{n} p_j d_{ji}[p - 2e_j] + 2A(p + 2e_i). \tag{5.4}$$

The latter, which can be used to raise the value of $\sum p_i$, contains n integrals since $d_{ii} = 0$.

Recurrence relations for a single p_i depend on the value of n. For $n = 3$ and i, j, k any permutation of 1, 2, 3, we have

$$\begin{aligned}(p_1 + p_2 + p_3 + 4)b_j b_k[p + 2e_i]& \\ + \{(p_i + p_j + 2)b_j d_{ki} + (p_i + p_k + 2)b_k d_{ji}\}[p]& \\ + p_i d_{ji} d_{ki}[p - 2e_i] = 2b_i A(p + 2e_j + 2e_k).&\end{aligned} \tag{5.5}$$

The analogous relation for $n = 4$ and i, j, k, m a permutation of 1, 2, 3, 4 is

$$\begin{aligned}(p_1 + p_2 + p_3 + p_4 + 6)b_j b_k b_m[p + 4e_i]& \\ + \sum(p_i + p_j + p_k + 4)b_j b_k d_{mi}[p + 2e_i]& \\ + \sum(p_i + p_j + 2)b_j d_{ki} d_{mi}[p] + p_i d_{ji} d_{ki} d_{mi}[p - 2e_i]& \\ = 2b_i^2 A(p + 2e_j + 2e_k + 2e_m),&\end{aligned} \tag{5.6}$$

where Σ denotes summation over cyclic permutations of j, k, m. This relation is especially useful if $\Sigma p_i = -6$, because the first term vanishes.

Equation (5.1) follows at once from the definition of $[p]$ and the identity

$$d_{ij} = b_j(a_i + b_i t) - b_i(a_j + b_j t). \tag{5.7}$$

To prove (5.3) we integrate both sides of

$$2\frac{d}{dt}\prod_{i=1}^{n}(a_i + b_i t)^{p_i/2} = \sum_{i=1}^{n} p_i b_i (a_i + b_i t)^{-1} \prod_{j=1}^{n}(a_j + b_j t)^{p_j/2} \tag{5.8}$$

with respect to t over the interval $[y, x]$.

If p is replaced by $p + 2e_i$, (5.3) becomes

$$(p_i + 2)b_i[p] + \sum_{\substack{j=1\\ j\neq i}}^{n} p_j b_j[p + 2e_i - 2e_j] = 2A(p + 2e_i), \tag{5.9}$$

and if p is replaced by $p - 2e_j$, (5.1) becomes

$$b_j[p + 2e_i - 2e_j] = b_i[p] - d_{ji}[p - 2e_j]. \tag{5.10}$$

Substitution of (5.10) in (5.9) yields (5.4). To prove (5.5) we use (5.7) twice to write $b_i^2(a_j + b_j t)(a_k + b_k t)$ as a quadratic polynomial in $a_i + b_i t$, multiply by $\Pi(a_r + b_r t)^{p_r/2}$, and integrate to get

$$\begin{aligned}b_i^2[p + 2e_j + 2e_k] = b_j b_k[p + 4e_i] + (b_j d_{ki} + b_k d_{ji})[p + 2e_i]& \\ + d_{ji} d_{ki}[p].&\end{aligned} \tag{5.11}$$

Next we replace p by $p + 2e_j + 2e_k$ in (5.3) with $n = 3$ and find

$$\begin{aligned}p_i b_i[p - 2e_i + 2e_j + 2e_k] + (p_j + 2)b_j[p + 2e_k]& \\ + (p_k + 2)b_k[p + 2e_j] = 2A(p + 2e_j + 2e_k).&\end{aligned} \tag{5.12}$$

In the first term we substitute (5.11) with p replaced by $p - 2e_i$; in the second and third terms we use (5.1) with or without replacement of j by k. The result is (5.5), and (5.6) has a similar proof starting from $b_i^3(a_j + b_j t)(a_k + b_k t)(a_m + b_m t)$ as a cubic polynomial in $a_i + b_i t$.

The following special cases of (5.1) show how to obtain (2.8), (2.9), (2.10), and (2.11) from (2.6) and (2.7):

$$d_{14}[-1, -1, -1, -3] = b_4[1, -1, -1, -3] - b_1[-1, -1, -1, -1], \tag{5.13}$$

$$(5.14)\qquad b_4[-1,-1,-1,-3,2] = d_{54}[-1,-1,-1,-3] + b_5[-1,-1,-1,-1],$$

$$(5.15)\qquad d_{34}[-1,-1,-3,-3] = b_4[-1,-1,-1,-3] - b_3[-1,-1,-3,-1],$$

$$(5.16)\qquad b_3[1,-1,-3,-3] = d_{13}[-1,-1,-3,-3] + b_1[-1,-1,-1,-3],$$

$$(5.17)\qquad b_3[1,1,-3,-3] = d_{23}[1,-1,-3,-3] + b_2[1,-1,-1,-3].$$

We have omitted $p_5 = 0$ in the two integrals on the right-hand side of (5.14). In (5.15), $[-1,-1,-3,-1]$ is found by interchanging the subscripts 3 and 4 in formula (2.8) specialized to $[-1,-1,-1,-3]$. Letting $[p] = [-1,-1,-1,-3]$ and $i = 4$ in (5.6), we get $[-1,-1,-1,-5]$ from $[-1,-1,-1,-3]$ and $[-1,-1,-1,-1]$, since the first term of (5.6) is 0. Equations (2.12) and (2.13) then follow from two more special cases of (5.1):

$$(5.18)\qquad b_4[1,-1,-1,-5] = d_{14}[-1,-1,-1,-5] + b_1[-1,-1,-1,-3],$$

$$(5.19)\qquad b_4[1,1,-1,-5] = d_{24}[1,-1,-1,-5] + b_2[1,-1,-1,-3].$$

The formulas resulting from this procedure can sometimes be simplified with the help of various identities:

$$(5.20)\qquad b_iX_j^2 - b_jX_i^2 = b_iY_j^2 - b_jY_i^2 = d_{ji},$$

$$(5.21)\qquad X_i^2Y_j^2 - Y_i^2X_j^2 = (x-y)d_{ji},$$

$$(5.22)\qquad \sum a_id_{jk} = \sum b_id_{jk} = \sum d_{im}d_{jk} = 0,$$

$$(5.23)\qquad \sum X_i^2d_{jk} = \sum Y_i^2d_{jk} = 0,$$

$$(5.24)\qquad \sum d_{ij}U_{ij}X_kY_k = 0,$$

where Σ denotes summation over cyclic permutations of i, j, k. These identities are obtained from definitions (2.1) to (2.3). Equation (5.22) is used to prove (5.23) and (5.23) to prove (5.24). Since R_D, unlike R_F, is symmetric in only its first two arguments, another useful relation is

$$(5.25)\qquad \begin{aligned} d_{1j}d_{ki}R_D\left(U_{1i}^2, U_{1j}^2, U_{1k}^2\right) &= d_{1k}d_{ij}R_D\left(U_{1i}^2, U_{1k}^2, U_{1j}^2\right) \\ &\quad + 3R_F\left(U_{12}^2, U_{13}^2, U_{14}^2\right) - \frac{3U_{1i}}{U_{1j}U_{1k}}, \end{aligned}$$

where i, j, k is any permutation of 2, 3, 4. This can be proved by using [5, (4.14)] to express both sides in terms of the symmetric functions R_G and R_F and simplifying with the help of (2.4).

The four cubic cases in Section 3 can be obtained from (5.1) and (5.4) as follows:

$$(5.26)\qquad b_3[3,-1,-3] = d_{13}[1,-1,-3] + b_1[1,-1,-1],$$

$$(5.27)\qquad b_2[3,-1,-1] = d_{12}[1,-1,-1] + b_1[1,1,-1],$$

$$(5.28)\qquad d_{12}[-3,-3,-3] = b_2[-1,-3,-3] - b_1[-3,-1,-3],$$

$$(5.29)\qquad 5b_1[1,1,1] = d_{21}[1,-1,1] + d_{31}[1,1,-1] + 2A(3,1,1).$$

Aside from permutation of indices, each integral on the right-hand side of these equations is among the 13 cubic cases listed in the first paragraph of Section 3. Equation (5.24) is replaced by two identities,

$$(5.30)\qquad \sum d_{ij}U_kX_kY_k = 0,$$

$$(5.31)\qquad b_iU_iX_jY_j - b_jU_jX_iY_i = d_{ji}U_k,$$

and (5.25) is replaced by

$$b_j d_{ki} R_D(U_i^2, U_j^2, U_k^2) = b_k d_{ij} R_D(U_i^2, U_k^2, U_j^2) + 3R_F(U_1^2, U_2^2, U_3^2) - \frac{3U_i}{U_j U_k}. \tag{5.32}$$

In these three equations i, j, k is any permutation of $1, 2, 3$, and $\sum$ denotes summation over cyclic permutations of i, j, k. Equation (5.23) is used to prove (5.30), and (5.20) to prove (5.31). Equation (5.32) is proved in the same way as (5.25) except that (3.2) is used in place of (2.4).

Department of Mathematics
Iowa State University
Ames, Iowa 50011

1. P. F. Byrd & M. D. Friedman, *Handbook of Elliptic Integrals for Engineers and Scientists*, 2nd ed., Springer-Verlag, New York, 1971.

2. B. C. Carlson, *Special Functions of Applied Mathematics*, Academic Press, New York, 1977.

3. B. C. Carlson, "Elliptic integrals of the first kind," *SIAM J. Math. Anal.*, v. 8, 1977, pp. 231–242.

4. B. C. Carlson, "Short proofs of three theorems on elliptic integrals," *SIAM J. Math. Anal.*, v. 9, 1978, pp. 524–528.

5. B. C. Carlson, "Computing elliptic integrals by duplication," *Numer. Math.*, v. 33, 1979, pp. 1–16.

6. B. C. Carlson & Elaine M. Notis, "ALGORITHM 577, Algorithms for incomplete elliptic integrals," *ACM Trans. Math. Software*, v. 7, 1981, pp. 398–403.

7. I. S. Gradshteyn & I. M. Ryzhik, *Table of Integrals, Series, and Products*, Academic Press, New York, 1980.

8. D. H. Lehmer, "The lemniscate constant," *Math. Comp.*, v. 3, 1948/49, pp. 550–551.

9. A. P. Prudnikov, Yu. A. Brychkov & O. I. Marichev, *Integrals and Series*, vol. 1, Gordon and Breach, New York, 1986.

10. John Todd, "The lemniscate constants," *Comm. ACM*, v. 18, 1975, pp. 14–19, 462.

11. D. G. Zill & B. C. Carlson, "Symmetric elliptic integrals of the third kind," *Math. Comp.*, v. 24, 1970, pp. 199–214.

MATHEMATICS OF COMPUTATION
VOLUME 49, NUMBER 180
OCTOBER 1987, PAGES S13-S17

Supplement to A Table of Elliptic Integrals of the Second Kind

By B. C. Carlson

This supplement contains Fortran routines for the standard functions $R_F(x,y,z)$ and $R_D(x,y,z)$, followed by two examples of their use in computing elliptic integrals.

```
C***********************************************************************
C
      DOUBLE PRECISION FUNCTION RF(X,Y,Z,ERRTOL,IERR)
C
C          THIS FUNCTION SUBROUTINE COMPUTES THE INCOMPLETE ELLIPTIC
C          INTEGRAL OF THE FIRST KIND,
C          RF(X,Y,Z) = INTEGRAL FROM ZERO TO INFINITY OF
C
C                                   -1/2      -1/2      -1/2
C                    (1/2)(T+X)    (T+Y)     (T+Z)     DT,
C
C          WHERE X, Y, AND Z ARE NONNEGATIVE AND AT MOST ONE OF THEM
C          IS ZERO.  IF ONE OF THEM IS ZERO, THE INTEGRAL IS COMPLETE.
C          THE DUPLICATION THEOREM IS ITERATED UNTIL THE VARIABLES ARE
C          NEARLY EQUAL, AND THE FUNCTION IS THEN EXPANDED IN TAYLOR
C          SERIES TO FIFTH ORDER.
C          REFERENCES: B. C. CARLSON AND E. M. NOTIS, ALGORITHMS FOR
C          INCOMPLETE ELLIPTIC INTEGRALS, ACM TRANSACTIONS ON MATHEMA-
C          TICAL SOFTWARE, 7 (1981), 398-403; B. C. CARLSON, COMPUTING
C          ELLIPTIC INTEGRALS BY DUPLICATION, NUMER. MATH., 33 (1979),
C          1-16.
C          AUTHORS: B. C. CARLSON AND ELAINE M. NOTIS, AMES LABORATORY-
C          DOE, IOWA STATE UNIVERSITY, AMES, IA 50011, AND R. L. PEXTON,
C          LAWRENCE LIVERMORE NATIONAL LABORATORY, LIVERMORE, CA 94550.
C          AUG. 1, 1979, REVISED JAN. 15, 1987.
C
C          CHECK VALUE: RF(0,1,2) = 1.31102 87771 46059 90523 24198
C          CHECK BY ADDITION THEOREM: RF(X,X+Z,X+W) + RF(Y,Y+Z,Y+W)
C          = RF(0,Z,W), WHERE X,Y,Z,W ARE POSITIVE AND X*Y = Z*W.
C
      INTEGER IERR,PRINTR
      DOUBLE PRECISION C1,C2,C3,E2,E3,EPSLON,ERRTOL,LAMDA
      DOUBLE PRECISION LOLIM,MU,S,UPLIM,X,XN,XNDEV,XNROOT
      DOUBLE PRECISION Y,YN,YNDEV,YNROOT,Z,ZN,ZNDEV,ZNROOT
C          INTRINSIC FUNCTIONS USED: DABS,DMAX1,DMIN1,DSQRT
C
C          PRINTR IS THE UNIT NUMBER OF THE PRINTER.
      DATA PRINTR/6/
C
```

©1987 American Mathematical Society
0025-5718/87 $1.00 + $.25 per page

```
C          LOLIM DETERMINES THE LOWER LIMIT AND UPLIM THE UPPER LIMIT
C          OF THE RANGE OF ADMISSIBLE VALUES OF X, Y, AND Z FOR WHICH
C          THE COMPUTATION WILL PROCEED WITHOUT UNDERFLOW OR OVERFLOW.
C          LOLIM IS NOT LESS THAN THE MACHINE MINIMUM MULTIPLIED BY 5.
C          UPLIM IS NOT GREATER THAN THE MACHINE MAXIMUM DIVIDED BY 5.
C
C          ACCEPTABLE VALUES FOR:   LOLIM        UPLIM
C          IBM 360/370 SERIES   :   3.0D-78      1.0D+75
C          CDC 6000/7000 SERIES :   1.0D-292     1.0D+321
C          UNIVAC 1100 SERIES   :   1.0D-307     1.0D+307
C          CRAY                 :   2.3D-2466    1.0D+2465
C          VAX 11 SERIES        :   1.5D-38      3.0D+37
C          IBM PC               :   1.5D-38      3.0D+37
C
C          WARNING: IF THIS PROGRAM IS CONVERTED TO SINGLE PRECISION,
C          THE VALUES FOR THE UNIVAC 1100 SERIES SHOULD BE CHANGED TO
C          LOLIM = 1.E-37 AND UPLIM = 1.E+37 BECAUSE THE MACHINE
C          EXTREMA CHANGE WITH THE PRECISION.
C
      DATA LOLIM/1.5D-38/, UPLIM/3.0D+37/
C
C          ON INPUT:
C
C          X, Y, AND Z ARE THE VARIABLES IN THE INTEGRAL RF(X,Y,Z).
C
C          ERRTOL IS CHOSEN TO DETERMINE THE ACCURACY OF THE COMPUTED
C          APPROXIMATION TO THE INTEGRAL.  TRUNCATION OF A TAYLOR SERIES
C          AFTER TERMS OF FIFTH ORDER INTRODUCES A RELATIVE ERROR LESS
C          THAN THE AMOUNT SHOWN IN THE SECOND COLUMN OF THE FOLLOWING
C          TABLE FOR EACH VALUE OF ERRTOL IN THE FIRST COLUMN.  IN ADDI-
C          TION TO THE TRUNCATION ERROR THERE WILL BE ROUNDOFF ERROR,
C          BUT IN PRACTICE THE TOTAL ERROR FROM BOTH SOURCES IS USUALLY
C          LESS THAN THE AMOUNT GIVEN IN THE TABLE.  SINCE THE TRUNCA-
C          TION ERROR IS LESS THAN (ERRTOL**6)/(4*(1-ERRTOL)), DECREAS-
C          ING ERRTOL BY A FACTOR OF 10 YIELDS SIX MORE DECIMAL DIGITS
C          OF ACCURACY AT THE EXPENSE OF ONE OR TWO MORE ITERATIONS OF
C          THE DUPLICATION THEOREM.
C
C          SAMPLE CHOICES:  ERRTOL    RELATIVE TRUNCATION
C                                     ERROR LESS THAN
C                           1.0D-3    3.0D-19
C                           3.0D-3    2.0D-16
C                           1.0D-2    3.0D-13
C                           3.0D-2    2.0D-10
C                           1.0D-1    3.0D-7
C
C          ON OUTPUT:
C
C          X, Y, Z, AND ERRTOL ARE UNALTERED.
C
C          IERR IS THE RETURN ERROR CODE:
C               IERR = 0 FOR NORMAL COMPLETION OF THE SUBROUTINE,
C               IERR = 1 FOR ABNORMAL TERMINATION.
C          ****************************************************************
C          WARNING: CHANGES IN THE PROGRAM MAY IMPROVE SPEED AT THE
C          EXPENSE OF ROBUSTNESS.
C
      IF (DMIN1(X,Y,Z) .LT. 0.D0) GO TO 100
      IF (DMIN1(X+Y,X+Z,Y+Z) .LT. LOLIM) GO TO 100
      IF (DMAX1(X,Y,Z) .LE. UPLIM) GO TO 112
  100 WRITE(PRINTR,104)
  104 FORMAT(1H0,42H*** ERROR - INVALID ARGUMENTS PASSED TO RF)
      WRITE(PRINTR,108) X,Y,Z
  108 FORMAT(1H ,4HX = ,D23.16,4X,4HY = ,D23.16,4X,4HZ = ,D23.16)
      IERR = 1
      GO TO 124
C
  112 IERR = 0
      XN = X
      YN = Y
      ZN = Z
C
  116 MU = (XN + YN + ZN) / 3.D0
      XNDEV = 2.D0 - (MU + XN) / MU
      YNDEV = 2.D0 - (MU + YN) / MU
      ZNDEV = 2.D0 - (MU + ZN) / MU
      EPSLON = DMAX1(DABS(XNDEV),DABS(YNDEV),DABS(ZNDEV))
      IF (EPSLON .LT. ERRTOL) GO TO 120
      XNROOT = DSQRT(XN)
      YNROOT = DSQRT(YN)
      ZNROOT = DSQRT(ZN)
      LAMDA = XNROOT * (YNROOT + ZNROOT) + YNROOT * ZNROOT
      XN = (XN + LAMDA) * 0.25D0
      YN = (YN + LAMDA) * 0.25D0
      ZN = (ZN + LAMDA) * 0.25D0
      GO TO 116
C
  120 C1 = 1.D0 / 24.D0
      C2 = 3.D0 / 44.D0
      C3 = 1.D0 / 14.D0
      E2 = XNDEV * YNDEV - ZNDEV * ZNDEV
      E3 = XNDEV * YNDEV * ZNDEV
      S = 1.D0 + (C1 * E2 - 0.1D0 - C2 * E3) * E2 + C3 * E3
      RF = S / DSQRT(MU)
C
  124 RETURN
      END
```

```
C***********************************************************************
C
      DOUBLE PRECISION FUNCTION RD(X,Y,Z,ERRTOL,IERR)
C
C          THIS FUNCTION SUBROUTINE COMPUTES AN INCOMPLETE ELLIPTIC
C          INTEGRAL OF THE SECOND KIND,
C          RD(X,Y,Z) = INTEGRAL FROM ZERO TO INFINITY OF
C
C                                    -1/2      -1/2      -3/2
C                      (3/2)(T+X)     (T+Y)     (T+Z)     DT,
C
C          WHERE X AND Y ARE NONNEGATIVE, X + Y IS POSITIVE, AND Z IS
C          POSITIVE.  IF X OR Y IS ZERO, THE INTEGRAL IS COMPLETE.
C          THE DUPLICATION THEOREM IS ITERATED UNTIL THE VARIABLES ARE
C          NEARLY EQUAL, AND THE FUNCTION IS THEN EXPANDED IN TAYLOR
C          SERIES TO FIFTH ORDER.
C          REFERENCE: B. C. CARLSON AND E. M. NOTIS, ALGORITHMS FOR
C          INCOMPLETE ELLIPTIC INTEGRALS, ACM TRANSACTIONS ON MATHEMA-
C          TICAL SOFTWARE, 7 (1981), 398-403; B. C. CARLSON, COMPUTING
C          ELLIPTIC INTEGRALS BY DUPLICATION, NUMER. MATH. 33 (1979),
C          1-16.
C          AUTHORS: B. C. CARLSON AND ELAINE M. NOTIS, AMES LABORATORY-
C          DOE, IOWA STATE UNIVERSITY, AMES, IA 50011, AND R. L. PEXTON,
C          LAWRENCE LIVERMORE NATIONAL LABORATORY, LIVERMORE, CA 94550.
C          AUG. 1, 1979, REVISED JAN. 15, 1987.
C
C          CHECK VALUE: RD(0,2,1) = 1.79721 03521 03388 31115 98837
C          CHECK: RD(X,Y,Z) + RD(Y,Z,X) + RD(Z,X,Y) = 3/DSQRT(X*Y*Z),
C          WHERE X, Y, AND Z ARE POSITIVE.
C
      INTEGER IERR,PRINTR
      DOUBLE PRECISION C1,C2,C3,C4,EA,EB,EC,ED,EF,EPSLON,ERRTOL,LAMDA
      DOUBLE PRECISION LOLIM,MU,POWER4,SIGMA,S1,S2,UPLIM,X,XN,XNDEV
      DOUBLE PRECISION XNROOT,Y,YN,YNDEV,YNROOT,Z,ZN,ZNDEV,ZNROOT
C          INTRINSIC FUNCTIONS USED: DABS,DMAX1,DMIN1,DSQRT
C
C          PRINTR IS THE UNIT NUMBER OF THE PRINTER.
      DATA PRINTR/6/
C
C          LOLIM DETERMINES THE LOWER LIMIT AND UPLIM THE UPPER LIMIT
C          OF THE RANGE OF ADMISSIBLE VALUES OF X, Y, AND Z FOR WHICH
C          THE COMPUTATION WILL PROCEED WITHOUT UNDERFLOW OR OVERFLOW.
C          LOLIM IS NOT LESS THAN 2/(MACHINE MAXIMUM)**(2/3).
C          UPLIM IS NOT GREATER THAN (0.1 * ERRTOL / MACHINE
C          MINIMUM) ** (2/3), WHERE ERRTOL IS DESCRIBED BELOW.
C          IN THE FOLLOWING TABLE IT IS ASSUMED THAT ERRTOL WILL
C          NEVER BE CHOSEN SMALLER THAN 1.D-5.
C
C          ACCEPTABLE VALUES FOR:   LOLIM          UPLIM
C          IBM 360/370 SERIES   :   6.0D-51        1.0D+48
C          CDC 6000/7000 SERIES :   5.0D-215       2.0D+191
C          UNIVAC 1100 SERIES   :   1.0D-205       2.0D+201
C          CRAY                 :   3.0D-1644      1.6D+1640
C          VAX 11 SERIES        :   1.0D-25        4.5D+21
C          IBM PC               :   1.0D-25        4.5D+21
C
C          WARNING: IF THIS PROGRAM IS CONVERTED TO SINGLE PRECISION,
C          THE VALUES FOR THE UNIVAC 1100 SERIES SHOULD BE CHANGED TO
C          LOLIM = 1.E-25 AND UPLIM = 2.E+21 BECAUSE THE MACHINE
C          EXTREMA CHANGE WITH THE PRECISION.
C
      DATA LOLIM/1.0D-25/, UPLIM/4.5D+21/
C
C          ON INPUT:
C
C          X, Y, AND Z ARE THE VARIABLES IN THE INTEGRAL RD(X,Y,Z).
C
C          ERRTOL IS CHOSEN TO DETERMINE THE ACCURACY OF THE COMPUTED
C          APPROXIMATION TO THE INTEGRAL.  TRUNCATION OF A TAYLOR SERIES
C          AFTER TERMS OF FIFTH ORDER INTRODUCES A RELATIVE ERROR LESS
C          THAN THE AMOUNT SHOWN IN THE SECOND COLUMN OF THE FOLLOWING
C          TABLE FOR EACH VALUE OF ERRTOL IN THE FIRST COLUMN.  IN ADDI-
C          TION TO THE TRUNCATION ERROR THERE WILL BE ROUNDOFF ERROR,
C          BUT IN PRACTICE THE TOTAL ERROR FROM BOTH SOURCES IS USUALLY
C          LESS THAN THE AMOUNT GIVEN IN THE TABLE.  SINCE THE TRUNCA-
C          TION ERROR IS LESS THAN 3 * ERRTOL ** 6 / (1-ERRTOL) ** 3/2,
C          DECREASING ERRTOL BY A FACTOR OF 10 YIELDS SIX MORE DECIMAL
C          DIGITS OF ACCURACY AT THE EXPENSE OF ONE OR TWO MORE ITERA-
C          TIONS OF THE DUPLICATION THEOREM.
C
C          SAMPLE CHOICES:  ERRTOL     RELATIVE TRUNCATION
C                                      ERROR LESS THAN
C                           1.0D-3     4.0D-18
C                           3.0D-3     3.0D-15
C                           1.0D-2     4.0D-12
C                           3.0D-2     3.0D-9
C                           1.0D-1     4.0D-6
C
C          ON OUTPUT:
C
C          X, Y, Z, AND ERRTOL ARE UNALTERED.
C
C          IERR IS THE RETURN ERROR CODE:
C               IERR = 0 FOR NORMAL COMPLETION OF THE SUBROUTINE,
C               IERR = 1 FOR ABNORMAL TERMINATION.
C
```

```
C         ****************************************************************
C         WARNING: CHANGES IN THE PROGRAM MAY IMPROVE SPEED AT THE
C         EXPENSE OF ROBUSTNESS.
C
      IF (DMIN1(X,Y) .LT. 0.D0) GO TO 100
      IF (DMIN1(X+Y,Z) .LT. LOLIM) GO TO 100
      IF (DMAX1(X,Y,Z) .LE. UPLIM) GO TO 112
  100 WRITE(PRINTR,104)
  104 FORMAT(1H0,42H*** ERROR - INVALID ARGUMENTS PASSED TO RD)
      WRITE(PRINTR,108) X,Y,Z
  108 FORMAT(1H ,4HX = ,D23.16,4X,4HY = ,D23.16,4X,4HZ = ,D23.16)
      IERR = 1
      GO TO 124
C
  112 IERR = 0
      XN = X
      YN = Y
      ZN = Z
      SIGMA = 0.D0
      POWER4 = 1.D0
C
  116 MU = (XN + YN + 3.D0 * ZN) * 0.2D0
      XNDEV = (MU - XN) / MU
      YNDEV = (MU - YN) / MU
      ZNDEV = (MU - ZN) / MU
      EPSLON = DMAX1(DABS(XNDEV),DABS(YNDEV),DABS(ZNDEV))
      IF (EPSLON .LT. ERRTOL) GO TO 120
      XNROOT = DSQRT(XN)
      YNROOT = DSQRT(YN)
      ZNROOT = DSQRT(ZN)
      LAMDA = XNROOT * (YNROOT + ZNROOT) + YNROOT * ZNROOT
      SIGMA = SIGMA + POWER4 / (ZNROOT * (ZN + LAMDA))
      POWER4 = POWER4 * 0.25D0
      XN = (XN + LAMDA) * 0.25D0
      YN = (YN + LAMDA) * 0.25D0
      ZN = (ZN + LAMDA) * 0.25D0
      GO TO 116
C
  120 C1 = 3.D0 / 14.D0
      C2 = 1.D0 / 6.D0
      C3 = 9.D0 / 22.D0
      C4 = 3.D0 / 26.D0
      EA = XNDEV * YNDEV
      EB = ZNDEV * ZNDEV
      EC = EA - EB
      ED = EA - 6.D0 * EB
      EF = ED + EC + EC
      S1 = ED * (- C1 + 0.25D0 * C3 * ED - 1.5D0 * C4 * ZNDEV * EF)
      S2 = ZNDEV * (C2 * EF + ZNDEV * (- C3 * EC + ZNDEV * C4 * EA))
      RD = 3.D0 * SIGMA + POWER4 * (1.D0 + S1 + S2) / (MU * DSQRT(MU))
C
  124 RETURN
      END
```

Example 1. The arc of Bernoulli's lemniscate $r^2 = \cos(2\theta)$ between the points (r,θ) and (ρ,ϕ), where $0 \le \theta \le \phi \le \pi/4$ and $0 \le \rho \le r \le 1$, has length [2,Ex.8.3-7]

(S.1) $$s = \int_\rho^r (1-u^4)^{-1/2}du.$$

Putting $u^2 = t$ and using (2.6) and (2.1)-(2.4), we find

$$s = \frac{1}{2}\int_{\rho^2}^{r^2}[t(1+t)(1-t)]^{-1/2}dt = R_F(U_{12}^2,U_{13}^2,U_{14}^2),$$

(S.2) $$(r^2-\rho^2)U_{14} = r(1-\rho^4)^{1/2} + \rho(1-r^4)^{1/2},$$

$$U_{12}^2 = U_{14}^2 + d_{13}d_{24} = U_{14}^2 + (-1)(-1) = U_{14}^2 + 1,$$

$$U_{13}^2 = U_{14}^2 + d_{12}d_{34} = U_{14}^2 + (-1)(1) = U_{14}^2 - 1.$$

We have chosen $a_1+b_1t = t$, $a_2+b_2t = 1+t$, $a_3+b_3t = 1-t$, and $a_4+b_4t = 1$. If $\rho = 1/\sqrt{3}$ and $r = 1/\sqrt{2}$, then $U_{14} = 7$ and

(S.3) $$s_1 = \int_{1/\sqrt{3}}^{1/\sqrt{2}}(1-u^4)^{-1/2}du = R_F(48,49,50) = 0.14286\ 30937\ 9176....$$

Here we have used the symmetry of R_F and the first Fortran code in this Supplement.

If $\rho = 0$, U_{14} reduces to $1/r$ and the arc length to

(S.4) $$\int_0^r (1-u^4)^{-1/2}du = R_F(r^{-2}-1,r^{-2},r^{-2}+1),$$

in agreement with [2,Ex.8.3-7]. The case $r = 1$, representing the length of a quadrant of the lemniscate [8],[10], is $R_F(0,1,2)$, with numerical value given to 25D in the comments of the Fortran code. A check on (S.3) is provided by splitting the integral into two parts,

$$s_2 = \int_0^{1/\sqrt{2}} (1-u^4)^{-1/2}du = R_F(1,2,3) = 0.72694\ 59354\ 6891\ldots, \tag{S.5}$$

$$s_3 = \int_0^{1/\sqrt{3}} (1-u^4)^{-1/2}du = R_F(2,3,4) = 0.58408\ 28416\ 7715\ldots. \tag{S.6}$$

We see that $s_1 = s_2 - s_3$ to 14D. The relation

$$R_F(1,2,3) - R_F(2,3,4) = R_F(48,49,50) \tag{S.7}$$

is a special case of the addition theorem (4.7).

Example 2. With the same notation and procedure as in Example 1, we find from (2.7) that

$$I = \int_\rho^r u^2(1-u^4)^{-1/2}du = \frac{1}{3} R_D(U_{14}^2-1, U_{14}^2+1, U_{14}^2) + r\rho/U_{14}, \tag{S.8}$$

where U_{14} is given in (S.2). A special case is

$$I_1 = \int_{1/\sqrt{3}}^{1/\sqrt{2}} u^2(1-u^4)^{-1/2}du = \frac{1}{3} R_D(48,50,49) + 1/7\sqrt{6} \tag{S.9}$$

$$= \frac{1}{3} \times 0.00291\ 57121\ 46567\ 96\ldots + 0.05832\ 11843\ 51980\ 43\ldots$$

$$= 0.05929\ 30884\ 00836\ 4\ldots,$$

where we have used the second Fortran code in this Supplement.

If $\rho = 0$, (S.8) reduces to

$$\int_0^r u^2(1-u^4)^{-1/2}du = \frac{1}{3} R_D(r^{-2}-1, r^{-2}+1, r^{-2}), \tag{S.10}$$

and the case $r = 1$ is the second lemniscate constant [8],[10],

$$\int_0^1 u^2(1-u^4)^{-1/2}du = \frac{1}{3} R_D(0,2,1) = 0.59907\ 01173\ 67796\ldots. \tag{S.11}$$

The value of $R_D(0,2,1)$ is given to 25D in the comments of the Fortran code, and a well-known relation between the two lemniscate constants [10,Theorem 2] takes the form $R_F(0,1,2)R_D(0,2,1) = 3\pi/4$. A check on (S.9) is provided by splitting the integral into two parts,

$$I_2 = \int_0^{1/\sqrt{2}} u^2(1-u^4)^{-1/2}du = \frac{1}{3} R_D(1,3,2) = 0.12505\ 74576\ 52385\ldots, \tag{S.12}$$

$$I_3 = \int_0^{1/\sqrt{3}} u^2(1-u^4)^{-1/2}du = \frac{1}{3} R_D(2,4,3) = 0.06576\ 43692\ 51548\ldots. \tag{S.13}$$

We see that $I_1 = I_2 - I_3$ to 13S. The relation

$$R_D(1,3,2) - R_D(2,4,3) = R_D(48,50,49) + \sqrt{6}/14 \tag{S.14}$$

is a special case of the addition theorem (4.13).